Minimization example for exam 17/2/2012

```
> restart:
  with(LinearAlgebra):
  with(plots) :
```

Define the function to minimize

```
> f := x+y+x*y ;
```

$$f := x + y + x\,y \tag{1}$$

Define the unilateral constraints

```
> g1 := x-y^2+1;
```

$$g1 := x - y^2 + 1 \tag{2}$$

```
> g2 := y-x^2+1 ;
```

$$g2 := y - x^2 + 1 \tag{3}$$

```
> solve(g1,{x}) ;
  solve(g2,{y}) ;
```

$$\{x = y^2 - 1\}$$

$$\{y = x^2 - 1\} \tag{4}$$

Plot the boundary of the constraints

```
> A := plot( [[t,t^2-1,t=-1.5..1.5],[t^2-1,t,t=-1.5..1.5]],scaling=
  CONSTRAINED) :
  B := plot( [[-1,-1],[0,-1],[-2/3,-5/9],[0,-1],[-1,-1],[-1,-2/3],
              [1/2-(1/2)*sqrt(5), 1/2-(1/2)*sqrt(5)]],style=point,
  symbolsize=30 ) :
  display(A,B) ;
```



Bouild the Lagrangian

```
> L := f - mu1 * g1 - mu2 * g2 ;
```

$$L := x + y + x\,y - \mu 1\,\left(x - y^2 + 1\right) - \mu 2\,\left(y - x^2 + 1\right) \tag{5}$$

First order KKT conditions

```
> EQ1 := diff( L, x ) ;
  EQ2 := diff( L, y ) ;
  EQ3 := mu1*g1 ;
  EQ4 := mu2*g2 ;
```

$$EQ1 := 1 + y - \mu 1 + 2\,\mu 2\,x$$

$$EQ2 := 1 + x + 2\,\mu 1\,y - \mu 2$$

$$EQ3 := \mu 1\,\left(x - y^2 + 1\right)$$

$$EQ4 := \mu 2\,\left(y - x^2 + 1\right) \tag{6}$$

```
> Hess := <<diff(L,x,x),diff(L,x,y)>|<diff(L,y,x),diff(L,y,y)>> ;
```

$$Hess := \begin{bmatrix} 2\,\mu 2 & 1 \\ 1 & 2\,\mu 1 \end{bmatrix} \tag{7}$$

Try mu1=0 and mu2 = 0

```
> EQS00 := subs(mu1=0,mu2=0,[EQ||(1..4)] ) ;
```

$$EQS00 := [1 + y, 1 + x, 0, 0] \quad (1.1)$$

```
> SOL00 := solve( EQS00, {x,y} ) ;
```

$$SOL00 := \{x = -1, y = -1\} \quad (1.2)$$

Check that the solution satisfy the constraints

```
> subs( SOL00, [g1,g2] ) ;
```

$$[-1, -1] \quad (1.3)$$

The solution is NOT feasible.

Try mu1=0 and mu2 free

```
> EQS00 := subs(mu1=0,[EQ||(1..4)] ) ;
```

$$EQS00 := [1 + y + 2\,\mu 2\, x, 1 + x - \mu 2, 0, \mu 2\,(y - x^2 + 1)] \quad (2.1)$$

```
> SOLy := solve( EQS00[4], {y} ) ;
```

$$SOLy := \{y = x^2 - 1\} \quad (2.2)$$

```
> EQS000 := subs( SOLy, EQS00 ) ;
```

$$EQS000 := [x^2 + 2\,\mu 2\, x, 1 + x - \mu 2, 0, 0] \quad (2.3)$$

```
> SOLmu2 := solve( EQS000[2], {mu2} ) ;
```

$$SOLmu2 := \{\mu 2 = 1 + x\} \quad (2.4)$$

```
> SOLx := solve( subs( SOLmu2, EQS000[1] ), {x} ) ;
```

$$SOLx := \{x = 0\}, \left\{x = -\frac{2}{3}\right\} \quad (2.5)$$

```
> SOL1 := op(SOLx[1]), op(subs( SOLx[1], SOLmu2 )), op(subs(SOLx
  [1],SOLy)), mu1 = 0 ;
```

$$SOL1 := x = 0, \mu 2 = 1, y = -1, \mu 1 = 0 \quad (2.6)$$

```
> SOL2 := op(SOLx[2]), op(subs( SOLx[2], SOLmu2 )), op(subs(SOLx
  [2],SOLy)), mu1 = 0 ;
```

$$SOL2 := x = -\frac{2}{3}, \mu 2 = \frac{1}{3}, y = -\frac{5}{9}, \mu 1 = 0 \quad (2.7)$$

Check that the solution satisfy the constraints

```
> subs( SOL1, [g1,g2] ) ;
  subs( SOL2, [g1,g2] ) ;
```

$$[0, 0]$$
$$\left[\frac{2}{81}, 0\right] \quad (2.8)$$

Ok, it is feasible, now check second order condition.
The active constraint are both g1 and g2 for first solution

```
> gradg1 := subs(SOL1,<diff(g1,x)|diff(g1,y)>) ;
```

$$gradg1 := \begin{bmatrix} 1 & 2 \end{bmatrix} \quad (2.9)$$

```
> gradg2 := subs(SOL2,<diff(g2,x)|diff(g2,y)>) ;
```

(2.10)

$$gradg2 := \begin{bmatrix} \frac{4}{3} & 1 \end{bmatrix} \quad (2.10)$$

The gradients are linearly independent so that Null space is empty

The active constraint for second point is only g2

```
> gradg2 := subs(SOL2,<diff(g2,x)|diff(g2,y)>) ;
```

$$gradg2 := \begin{bmatrix} \frac{4}{3} & 1 \end{bmatrix} \quad (2.11)$$

```
> N := op(NullSpace(gradg2)) ;
```

$$N := \begin{bmatrix} -\frac{3}{4} \\ 1 \end{bmatrix} \quad (2.12)$$

Check if the hessian is SPD in the kernel

```
> Hess00 := subs(SOL2,Hess) ;
```

$$Hess00 := \begin{bmatrix} \frac{2}{3} & 1 \\ 1 & 0 \end{bmatrix} \quad (2.13)$$

```
> Transpose(N).Hess00.N ; subs(mu1=0,mu2=0,%) ;
```

$$-\frac{9}{8}$$

$$-\frac{9}{8} \quad (2.14)$$

It is not a minimum point!

Try mu2=0 and mu1 free

```
> EQS00 := subs(mu2=0,[EQ||(1..4)] ) ;
```

$$EQS00 := \left[1 + y - \mu 1, 1 + x + 2\,\mu 1\, y, \mu 1\,\left(x - y^2 + 1\right), 0\right] \quad (3.1)$$

```
> SOLx := solve( EQS00[2], {x} ) ;
```

$$SOLx := \{x = -1 - 2\,\mu 1\, y\} \quad (3.2)$$

```
> EQS000 := subs( SOLx, EQS00 ) ;
```

$$EQS000 := \left[1 + y - \mu 1, 0, \mu 1\,\left(-2\,\mu 1\, y - y^2\right), 0\right] \quad (3.3)$$

```
> SOLmu1 := solve( EQS000[1], {mu1} ) ;
```

$$SOLmu1 := \{\mu 1 = 1 + y\} \quad (3.4)$$

```
> SOLy := solve( subs( SOLmu1, EQS000[3] ), {y} ) ;
```

$$SOLy := \{y = 0\}, \{y = -1\}, \left\{y = -\frac{2}{3}\right\} \quad (3.5)$$

```
> SOL1 := op(SOLy[1]), op(subs( SOLy[1], SOLmu1 )), op(subs(SOLy
[1],SOLx)), mu2 = 0 ;
```

$$SOL1 := y = 0, \mu 1 = 1, x = -1, \mu 2 = 0 \quad (3.6)$$

```
> SOL2 := op(SOLy[2]), op(subs( SOLy[2], SOLmu1 )), op(subs(SOLy[2],subs( SOLy[2], SOLmu1 ),SOLx)), mu2 = 0 ;
```

$$SOL2 := y = -1, \mu 1 = 0, x = -1, \mu 2 = 0 \tag{3.7}$$

```
> SOL3 := op(SOLy[3]), op(subs( SOLy[3], SOLmu1 )), op(subs(SOLy[3],subs( SOLy[2], SOLmu1 ),SOLx)), mu2 = 0 ;
```

$$SOL3 := y = -\frac{2}{3}, \mu 1 = \frac{1}{3}, x = -1, \mu 2 = 0 \tag{3.8}$$

Check that the solution satisfy the constraints

```
> subs( SOL1, [g1,g2] ) ;
  subs( SOL2, [g1,g2] ) ;
  subs( SOL3, [g1,g2] ) ;
```

$$[0, 0]$$
$$[-1, -1]$$
$$\left[-\frac{4}{9}, -\frac{2}{3}\right] \tag{3.9}$$

Only one point is feasible
The active constraint are both g1 and g2 for first solution

```
> gradg1 := subs(SOL1,<diff(g1,x)|diff(g1,y)>) ;
```

$$gradg1 := \begin{bmatrix} 1 & 0 \end{bmatrix} \tag{3.10}$$

```
> gradg2 := subs(SOL1,<diff(g2,x)|diff(g2,y)>) ;
```

$$gradg2 := \begin{bmatrix} 2 & 1 \end{bmatrix} \tag{3.11}$$

The gradients are linearly independent so that Null space is empty

Try mu1 and mu2 free

```
> EQ3bis := EQ3/mu1; EQ4bis := EQ4/mu2;
```

$$EQ3bis := x - y^2 + 1$$
$$EQ4bis := y - x^2 + 1 \tag{4.1}$$

```
> subs( solve( EQ3bis, {x} ), EQ4bis ) ; SOLy := solve(%,{y}) ;
```

$$y - (y^2 - 1)^2 + 1$$
$$SOLy := \{y = 0\}, \{y = -1\}, \left\{y = \frac{1}{2} - \frac{1}{2}\sqrt{5}\right\}, \left\{y = \frac{1}{2}\sqrt{5} + \frac{1}{2}\right\} \tag{4.2}$$

```
> SOLx := solve( EQ3bis, {x} ) ;
```

$$SOLx := \{x = y^2 - 1\} \tag{4.3}$$

```
> SOL1 := op(subs( SOLy[1],SOLx)), op(SOLy[1]) ;
```

$$SOL1 := x = -1, y = 0 \tag{4.4}$$

```
> subs(SOL1,{EQ1,EQ2}) ; SOL1 := SOL1, op(solve( %, {mu1,mu2})) ;
```

$$\{-\mu 2, 1 - \mu 1 - 2\mu 2\}$$
$$SOL1 := x = -1, y = 0, \mu 1 = 1, \mu 2 = 0 \tag{4.5}$$

```
> SOL2 := op(subs( SOLy[2],SOLx)), op(SOLy[2]) ;
```

$$SOL2 := x = 0, y = -1 \tag{4.6}$$

```
> subs(SOL2,{EQ1,EQ2}) ; SOL2 := SOL2, op(solve( %, {mu1,mu2})) ;
```
$$\{-\mu 1, 1 - 2\,\mu 1 - \mu 2\}$$
$$SOL2 := x = 0, y = -1, \mu 1 = 0, \mu 2 = 1 \quad (4.7)$$
```
> SOL3 := simplify(op(subs( SOLy[3],SOLx))), op(SOLy[3]) ;
```
$$SOL3 := x = \frac{1}{2} - \frac{1}{2}\sqrt{5}, y = \frac{1}{2} - \frac{1}{2}\sqrt{5} \quad (4.8)$$
```
> subs(SOL3,{EQ1,EQ2}) ; SOL3 := SOL3, op(solve( %, {mu1,mu2})) ;
```
$$\left\{\frac{3}{2} - \frac{1}{2}\sqrt{5} - \mu 1 + 2\,\mu 2\left(\frac{1}{2} - \frac{1}{2}\sqrt{5}\right), \frac{3}{2} - \frac{1}{2}\sqrt{5} + 2\,\mu 1\left(\frac{1}{2} - \frac{1}{2}\sqrt{5}\right) - \mu 2\right\}$$
$$SOL3 := x = \frac{1}{2} - \frac{1}{2}\sqrt{5}, y = \frac{1}{2} - \frac{1}{2}\sqrt{5}, \mu 1 = -\frac{1}{2} + \frac{3}{10}\sqrt{5}, \mu 2 = -\frac{1}{2} + \frac{3}{10}\sqrt{5} \quad (4.9)$$
```
> SOL4 := simplify(op(subs( SOLy[4],SOLx))), op(SOLy[4]) ;
```
$$SOL4 := x = \frac{1}{2}\sqrt{5} + \frac{1}{2}, y = \frac{1}{2}\sqrt{5} + \frac{1}{2} \quad (4.10)$$
```
> subs(SOL4,{EQ1,EQ2}) ; SOL4 := SOL3, op(solve( %, {mu1,mu2})) ;
```
$$\left\{\frac{3}{2} + \frac{1}{2}\sqrt{5} - \mu 1 + 2\,\mu 2\left(\frac{1}{2}\sqrt{5} + \frac{1}{2}\right), \frac{3}{2} + \frac{1}{2}\sqrt{5} + 2\,\mu 1\left(\frac{1}{2}\sqrt{5} + \frac{1}{2}\right) - \mu 2\right\}$$
$$SOL4 := x = \frac{1}{2} - \frac{1}{2}\sqrt{5}, y = \frac{1}{2} - \frac{1}{2}\sqrt{5}, \mu 1 = -\frac{1}{2} + \frac{3}{10}\sqrt{5}, \mu 2 = -\frac{1}{2} + \frac{3}{10}\sqrt{5}, \mu 1 = -\frac{3}{10}\sqrt{5} - \frac{1}{2}, \mu 2 = -\frac{3}{10}\sqrt{5} - \frac{1}{2} \quad (4.11)$$

Solution 1 and 2 already considered

```
> evalf( [SOL3]) ;
  evalf( [SOL4]) ;
```
$$[x = -0.6180339880, y = -0.6180339880, \mu 1 = 0.1708203931, \mu 2 = 0.1708203931]$$
$$[x = -0.6180339880, y = -0.6180339880, \mu 1 = 0.1708203931, \mu 2 = 0.1708203931, \mu 1 = -1.170820393, \mu 2 = -1.170820393] \quad (4.12)$$

Solution 3 is ok, 4 have negative multiplyer and must be discarded

Check that the solution satisfy the constraints

```
> simplify(subs( SOL3, [g1,g2] )) ;
```
$$[0, 0] \quad (4.13)$$

Ok, it is feasible, now check second order condition. The active constraint is only g2 ==> find the kernel of the gradient of g2

```
> gradg1 := subs(SOL3,<diff(g1,x)|diff(g1,y)>) ;
  gradg2 := subs(SOL3,<diff(g2,x)|diff(g2,y)>) ;
```
$$gradg1 := \left[\begin{array}{cc} 1 & \sqrt{5} - 1 \end{array}\right]$$
$$(4.14)$$

$$gradg2 := \begin{bmatrix} \sqrt{5}-1 & 1 \end{bmatrix} \tag{4.14}$$

```
> K := <gradg1,gradg2> ;
```

$$K := \begin{bmatrix} 1 & \sqrt{5}-1 \\ \sqrt{5}-1 & 1 \end{bmatrix} \tag{4.15}$$

Find the null space of gradg2

```
>  N := op(NullSpace(K)) ;
```

$$N := \tag{4.16}$$